



ΑΝΟΙΚΤΑ ακαδημαϊκά
μαθήματα **ΠΠ**



Εισαγωγή στους Υπολογιστές


Ενότητα #3: Logical IF Statements, Lookup Tables, Linear Interpolation

Καθ. Δημήτρης Ματαράς
Πολυτεχνική Σχολή
Τμήμα Χημικών Μηχανικών

Logical **IF** Statements, **Lookup Tables**, **Linear Interpolation**

BASIC ENGINEERING SERIES AND TOOLS

ENGINEERING COMPUTATIONS
AN INTRODUCTION
USING MATLAB® AND EXCEL®



JOSEPH C. MUSTO • WILLIAM E. HOWARD • RICHARD R. WILLIAMS


A. Καλαμπούνιας

The **IF** Statement

- Οι *λογικές εκφράσεις* (*Logical statements*) χρησιμοποιούνται για να ελεγχθεί η ροή των υπολογισμών κάποιου προγράμματος ή ενός λογιστικού φύλλου (spreadsheet)
- Το MATLAB και το Excel έχουν μεταξύ των άλλων εντολών και την εντολή **IF** για τον έλεγχο λογικών εκφράσεων
- Με το **IF** ελέγχουμε μια συνθήκη και η ροή του προγράμματος εξαρτάται από το αν είναι αληθής (true) ή ψευδής (false)

Παράδειγμα ελέγχου λογικής έκφρασης

- Αναζητούμε την απόλυτη τιμή ενός αριθμού χρησιμοποιώντας το Excel και χωρίς να χρησιμοποιήσουμε την αντίστοιχη εγγενή συνάρτηση ABS.
- Η διαδικασία επίλυσης μπορεί να περιγραφεί ως εξής:
“εάν ο αριθμός είναι μικρότερος από το μηδέν, τότε παίρνουμε το αρνητικό αυτού, αλλιώς αφήνουμε τον αριθμό ως έχει”
- Αυτή η *if-then-else* λογική χρησιμοποιείται εκτενώς στις γλώσσες προγραμματισμού και υπάρχει στο Excel και στο MATLAB.
- Format of Excel **IF** statement:
-if(logical_test;value_if_true;value_if_false)



The screenshot shows the Microsoft Excel interface. The formula bar at the top displays the formula `=IF(A1<0,-A1;A1)`. Below the formula bar, a grid of cells is visible. The grid has columns labeled A through G and rows labeled 1 through 7. The data in the grid is as follows:

	A	B	C	D	E	F	G
1	-1	1					
2	10	10					
3	25	25					
4	-45	45					
5	15	15					
6							
7							

Excel **IF** Statement

- Το IF είναι μια συνάρτηση που έχει τρία ορίσματα
- Το **πρώτο όρισμα** είναι η λογική έκφραση που πρόκειται να ελεγχθεί.
- Το **δεύτερο όρισμα** είναι η ενέργεια που πρέπει να γίνει αν η συνθήκη είναι αληθής.
- Το **τρίτο όρισμα** είναι η ενέργεια που πρέπει να γίνει αν η συνθήκη είναι ψευδής.

<i>αύξων αριθμός</i>	<i>Όνοματεπώνυμο</i>	<i>Έτος</i>	<i>Βαθμός</i>	<i>χαρακτηρισμός</i>
1	Κώστας	2	2	αποτυχία
2	Γιώργος	2	4	αποτυχία
3	Ιωάννα	2	1	αποτυχία
4	Ελένη	2	3	αποτυχία
5	Διονύσης	2	4	αποτυχία
6	Βάλια	2	6	επιτυχία
7	Μαρία	2	7	επιτυχία
8	Χαρά	2	8	επιτυχία
9	Δημήτρης	2	9	επιτυχία
10	Χρήστος	2	5	επιτυχία

Χρησιμοποιώντας την παρακάτω formula μπορούμε να υπολογίσουμε το χαρακτηρισμό στον παρακάτω πίνακα

`=IF(D2>=5;"επιτυχία";"αποτυχία")`

Nesting IF Statements (πολλαπλοί εσωτερικοί έλεγχοι)

- Στο MATLAB υπάρχει η δομή *if-then-else*
- Στο Excel πρέπει να χρησιμοποιηθούν πολλαπλοί εσωτερικοί έλεγχοι (*nest IF statements*)

Παράδειγμα:

Θέλουμε να μετατρέψουμε την αριθμητική βαθμολογία σε χαρακτηρισμούς A, B, C, D, F με βάση την παρακάτω κλίμακα:

- Μεγαλύτερο από ή ίσο με 90 = A,
- Μεγαλύτερο από ή ίσο με 80 και μικρότερο από 90 = B
- Μεγαλύτερο από ή ίσο με 70 και μικρότερο από 80 = C
- Μεγαλύτερο από ή ίσο με 60 και μικρότερο από 70 = D
- Μικρότερο από ή ίσο με 60 = F

Excel Formula

	A	B	C	D	E	F	G
1	55	=IF(A1>=90,"A",IF(A1>=80,"B",IF(A1>=70,"C",IF(A1>=60,"D","F"))))					
2							

(ξεκινώντας τον έλεγχο των βαθμών από το μεγαλύτερο προς το μικρότερο)
Εάν η πρόταση IF είναι αληθής, τότε στο B1 κελί μπαίνει το “A”

Εάν η πρόταση IF δεν είναι αληθής, τότε προχωράμε στην επόμενη πρόταση IF

Προσοχή στη χρήση του “>=” για το μεγαλύτερο ίσο από:
Εάν είχε γραφτεί μόνο “>” (μεγαλύτερο από), τότε ένας βαθμός ίσος με 90 ακριβώς θα αντιστοιχούσε στο B.

παράδειγματα

	A	B
1	92	A
2		

	A	B	
1	76	C	
2			

	A	B
1	90	A
2		

	A	B	
1	60	D	
2			

	A	B
1	89.999	B
2		

	A	B
1	55	F
2		

	A	B
1	88	B
2		

Εναλλακτική λύση #1

	A	B	C	D	E	F	G
1	89	=IF(A1<60,"F",IF(A1<70,"D",IF(A1<80,"C",IF(A1<90,"B","A"))))					
2		IF(logical_test, [value_if_true], [value_if_false])					

(ξεκινώντας τον έλεγχο των βαθμών από το μικρότερο προς το μεγαλύτερο)

Εναλλακτική λύση #2

A	B	C	D	E	F	G	
60	=IF(A1>=60,IF(A1>=70,IF(A1>=80,IF(A1>=90,"A","B"),"C"),"D"),"F")						

IF(logical_test, [value_if_true], [value_if_false])

First Logical Test


Value if True (leads to further tests)

Value if False

Προσοχή στις παρενθέσεις!

Παράδειγμα εφαρμογής του **if**

Copyright © The McGraw-Hill Companies, Inc. Permission required for reproduction or display.




Θα χρησιμοποιηθεί η δομή **if** για να διερευνηθεί η αποδοχή ή η απόρριψη μετρήσεων της διαμέτρου των οπών που φαίνονται στο παραπάνω σχήμα.

Η ιδανική τιμή για αποδεκτή τιμή της διαμέτρου είναι 0.2500 inch.

- Το **άνω όριο** είναι: $0.2500 + 0.0010$
- Το **κάτω όριο** είναι: $0.2500 - 0.0005$

Άρα, οι αποδεκτές τιμές της διαμέτρου πρέπει να κυμαίνονται εντός του διαστήματος: 0.2495 – 0.2510 inches.

Εύρος αποδεκτών τιμών της διαμέτρου: 0.2495 – 0.2510 inch.



Copyright © The McGraw-Hill Companies, Inc. Permission required for reproduction or display.

	A	B
1	Target Diameter (in)	0.2500
2	High Tolerance (in)	0.0010
3	Low Tolerance(in)	-0.0005
4		
5	Measured Diameter (in)	Tolerance Check
6	0.2501	ACCEPT
7	0.2496	ACCEPT
8	0.2498	ACCEPT
9	0.2512	REJECT HIGH
10	0.2502	ACCEPT
11	0.2512	REJECT HIGH
12	0.2495	ACCEPT
13	0.2499	ACCEPT
14	0.2500	ACCEPT
15	0.2493	REJECT LOW
16	0.2496	ACCEPT
17	0.2506	ACCEPT
18	0.2502	ACCEPT
19	0.2499	ACCEPT
20	0.2506	ACCEPT
21	0.2503	ACCEPT
22	0.2496	ACCEPT
23	0.2494	REJECT LOW
24	0.2497	ACCEPT
25	0.2508	ACCEPT

Στο κελί B6 γράφουμε:
`=IF((A6-B1)>B2,"REJECT HIGH",IF((A6-B1)<B3,"REJECT LOW","ACCEPT"))`

Η πρόταση COUNTIF(range, criteria)

Με την εντολή **countif** μετρώ το πλήθος των accepted, rejected high, rejected low μετρήσεων

Στο κελί E2 γράφω:

=COUNTIF(B6:B25,"ACCEPT")

Στο κελί E3 γράφω:

=COUNTIF(B6:B25,"REJECT HIGH")

Στο κελί E4 γράφω:

=COUNTIF(B6:B25,"REJECT LOW")

Στο κελί F2 γράφω:

=E2/SUM(\$E\$2:\$E\$4) και κάνω μορφοποίηση να μου το δείχνει ως ποσοστό

Copyright © The McGraw-Hill Companies, Inc. Permission required for reproduction or display.

D	E	F
	Number	Percent
Accepted	16	80%
Rejected High	2	10%
Rejected Low	2	10%

Copyright © The McGraw-Hill Companies, Inc. Permission required for reproduction or display.

	A	B
1	Target Diameter (in)	0.2500
2	High Tolerance (in)	0.0010
3	Low Tolerance(in)	-0.0005
4		
5	Measured Diameter (in)	Tolerance Check
6	0.2501	ACCEPT
7	0.2496	ACCEPT
8	0.2498	ACCEPT
9	0.2512	REJECT HIGH
10	0.2502	ACCEPT
11	0.2512	REJECT HIGH
12	0.2495	ACCEPT
13	0.2499	ACCEPT
14	0.2500	ACCEPT
15	0.2493	REJECT LOW
16	0.2496	ACCEPT
17	0.2506	ACCEPT
18	0.2502	ACCEPT
19	0.2499	ACCEPT
20	0.2506	ACCEPT
21	0.2503	ACCEPT
22	0.2496	ACCEPT
23	0.2494	REJECT LOW
24	0.2497	ACCEPT
25	0.2508	ACCEPT

Τελική μορφή του spreadsheet

Copyright © The McGraw-Hill Companies, Inc. Permission required for reproduction or display.

	A	B	C	D	E	F
1	Target Diameter (in)	0.2500			Number	Percent
2	High Tolerance (in)	0.0010		Accepted	16	80%
3	Low Tolerance(in)	-0.0005		Rejected High	2	10%
4				Rejected Low	2	10%
5	Measured Diameter (in)	Tolerance Check				
6	0.2501	ACCEPT				
7	0.2496	ACCEPT				
8	0.2498	ACCEPT				
9	0.2512	REJECT HIGH				
10	0.2502	ACCEPT				
11	0.2512	REJECT HIGH				
12	0.2495	ACCEPT				
13	0.2499	ACCEPT				
14	0.2500	ACCEPT				
15	0.2493	REJECT LOW				
16	0.2496	ACCEPT				
17	0.2506	ACCEPT				
18	0.2502	ACCEPT				
19	0.2499	ACCEPT				
20	0.2506	ACCEPT				
21	0.2503	ACCEPT				
22	0.2496	ACCEPT				
23	0.2494	REJECT LOW				
24	0.2497	ACCEPT				
25	0.2508	ACCEPT				

Συνδυασμένες συνθήκες

- Οι εκφράσεις **AND** και **OR** μπορούν να χρησιμοποιηθούν για εξέταση πολλαπλών συνθηκών
- **Formats:**
 - AND(condition1, condition2...) True μόνο όταν όλες οι συνθήκες είναι αληθείς
 - OR(condition1, condition2...) True μόνο όταν μια από τις συνθήκες είναι αληθής.

Παράδειγμα

Έχουμε 5 διαφορετικά αντικείμενα, τα οποία περνούν ποιοτικό έλεγχο εξετάζοντας κάποια τεστ. Κάθε τεστ παίρνει χαρακτηρισμό Pass (P) ή Failed (F). Τα αποτελέσματα των τεστ παρουσιάζονται στον παρακάτω πίνακα.

Εξετάζουμε στη συνέχεια τα εξής:

- 1) Το αντικείμενο είναι αποδεκτό όταν περνά και τα 3 τεστ.
- 2) Το αντικείμενο είναι αποδεκτό όταν περνά κάποιο από τα τεστ.

Copyright © The McGraw-Hill Companies, Inc. Permission required for reproduction or display.

	A	B	C	D
1	Part Number	Test 1	Test 2	Test 3
2	1	P	F	F
3	2	P	F	P
4	3	P	P	P
5	4	F	F	F
6	5	P	P	P
-				

Έλεγχος του 1^{ου} κριτηρίου:

Στο κελί E2 γράφω: =IF(AND(B2="P",C2="P",D2="P"),"ACCEPT","REJECT")

Κάνω click & drag από το κελί E3:E6.

Έλεγχος του 2^{ου} κριτηρίου:

Στο κελί F2 γράφω: =IF(OR(B2="P",C2="P",D2="P"),"ACCEPT","REJECT")

Κάνω click & drag από το κελί F3:F6.

Copyright © The McGraw-Hill Companies, Inc. Permission required for reproduction or display.

	A	B	C	D	E	F
1	Part Number	Test 1	Test 2	Test 3	Criterion 1	Criterion 2
2	1	P	F	F	Reject	Accept
3	2	P	F	P	Reject	Accept
4	3	P	P	P	Accept	Accept
5	4	F	F	F	Reject	Reject
6	5	P	P	P	Accept	Accept

Lookup Tables

- Συχνά χρειάζεται να καλούμε δεδομένα, τα οποία είναι αποθηκευμένα σε κάποιο πίνακα
- Για παράδειγμα, ας θεωρήσουμε τα παρακάτω μέταλλα και τις ιδιότητες τους.

Metal	Modulus of Elasticity, psi	Density, lb/in ³	Yield Strength, psi
Aluminum 2014-T6	10,600,000	0.101	60,000
Aluminum 6061-T6	10,000,000	0.098	37,000
Stainless Steel 304	28,000,000	0.284	30,000
Structural Steel A36	29,000,000	0.284	36,000

Παράδειγμα **Lookup Table**

- Ένας μηχανικός σε μια βιομηχανία πραγματοποιεί υπολογισμούς χρησιμοποιώντας τις ιδιότητες των προηγούμενων τεσσάρων υλικών
- Αντί λοιπόν ο μηχανικός να αναζητά και να γράφει αυτές τις ιδιότητες κάθε φορά, το Excel μπορεί να χρησιμοποιήσει ένα **lookup table** για να αποθηκεύσει αυτές τις ιδιότητες και να επιλέγει τις σωστές τιμές για δεδομένο υλικό

Παράδειγμα **Lookup Table**

- Σε αυτό το παράδειγμα, καθώς οι ιδιότητες είναι υπό μορφή στηλών, ο πίνακας λέγεται (κατακόρυφος) **vertical lookup table**
- Εάν τα δεδομένα δίνονται υπό μορφή γραμμών, ο πίνακας λέγεται (οριζόντιος) **horizontal lookup table**
- Η εντολή-σύνταξη για τον **vertical lookup table** στο Excel είναι:
VLOOKUP(lookup value, table range, column number, true/false)

Vertical Lookup Table Command

VLOOKUP(lookup value, table range, column number, true/false)

- Η **Lookup value** είναι το κελί που περιέχει το στοιχείο αναγνώρισης (identifier) για τα δεδομένα που πρόκειται να ελεγχθούν (data to be looked up). Στην περίπτωσή μας, το κελί θα περιέχει το όνομα του υλικού
- **Table range** είναι ένα group από κελιά που περιέχουν τον lookup table
- **Column number** είναι η στήλη που περιέχει την επιθυμητή ιδιότητα του υλικού
- Το **true/false** όρισμα είναι προαιρετικό: Το **true** επιτρέπει μια προσεγγιστική ταύτιση, ενώ το **false** απαιτεί μια πλήρη ταύτιση

Παράδειγμα **VLOOKUP**

- Εδώ παρουσιάζονται τα δεδομένα που έχουν εισαχθεί στο Excel
- Πολλές φορές είναι βολικό να εισαχθούν τα δεδομένα του πίνακα σε ένα ξεχωριστό sheet του workbook
- Σημειώνεται ότι έχουμε δώσει σε κάθε υλικό ένα σύντομο όνομα "**Short Name**" και έτσι δεν χρειάζεται να δίνουμε το πλήρες όνομα κάθε φορά

	A	B	C	D	E
1					
2					
3					
4					
5					
6					
7					
8	Short Name	Long Name	Modulus	Density	Yield Strength
9	A36	Structural Steel A36	29000000	0.284	36000
10	2014	Aluminum 2014-T6	10600000	0.101	60000
11	6061	Aluminum 6061-T6	10000000	0.098	37000
12	SS	Stainless Steel 304	28000000	0.284	30000

Παράδειγμα **VLOOKUP**

- Το σύντομο όνομα έχει εισαχθεί στο κελί cell B1. Θέλουμε να έχουμε παρούσα την υπόλοιπη πληροφορία αυτόματα

	A	B	C	D	E
1	Short Name	A36			
2					
3	Long Name				
4	Modulus				
5	Density				
6	Yield Strength				
7					
8	Short Name	Long Name	Modulus	Density	Yield Strength
9	A36	Structural Steel A36	29000000	0.284	36000
10	2014	Aluminum 2014-T6	10600000	0.101	60000
11	6061	Aluminum 6061-T6	10000000	0.098	37000
12	SS	Stainless Steel 304	28000000	0.284	30000

Παράδειγμα **VLOOKUP**

- Το πρώτο όρισμα είναι η διεύθυνση του κελιού που περιέχει το σύντομο όνομα του υλικού που πρόκειται να αναζητηθεί

	A	B	C	D
1	Short Name	A36		
2				
3	Long Name	=vlookup(B1	I	
4	Modulus			

VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

Παράδειγμα **VLOOKUP**

- Το δεύτερο όρισμα είναι το εύρος του κελιού που περιέχει τον lookup table. Η πρώτη στήλη του highlighted διαστήματος πρέπει να περιέχει την ανεξάρτητη μεταβλητή (the Short Name)

	A	B	C	D	E
1	Short Name	A36			
2					
3	Long Name	=vlookup(B1,A8:E12			
4	Modulus	VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])			
5	Density				
6	Yield Strength				
7					
8	Short Name	Long Name	Modulus	Density	Yield Strength
9	A36	Structural Steel A36	29000000	0.284	36000
10	2014	Aluminum 2014-T6	10600000	0.101	60000
11	6061	Aluminum 6061-T6	10000000	0.098	37000
12	SS	Stainless Steel 304	28000000	0.284	30000

Παράδειγμα **VLOOKUP**

- Το τρίτο όρισμα περιέχει τον αριθμό στήλης που περιέχει την επιθυμητή ιδιότητα. Υπενθυμίζεται ότι η στήλη που περιέχει το short name είναι η στήλη 1. Άρα, το long name είναι η στήλη 2.

	A	B	
1	Short Name	A36	
2			
3	Long	=vlookup(B1,A8:E12,2)	
4	Modulus		

Παράδειγμα **VLOOKUP**

- Αποτέλεσμα:

	A	B	C
1	Short Name	A36	
2			
3	Long Name	Structural Steel A36	
4	Modulus		
5	Density		

- Κοιτώντας τη διεύθυνση του κελιού, αυτή η εξίσωση μπορεί να αντιγραφεί σε άλλα κελιά...

	A	B	C
1	Short Name	A36	
2			
3	Long	=VLOOKUP(\$B\$1,\$A\$8:\$E\$12,2)	
4	Modulus		
5	Density		

VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

κρατάω τις συντεταγμένες σταθερές!

Παράδειγμα **VLOOKUP**

- Αλλάζοντας μόνο τον αριθμό της στήλης, παίρνουμε την επιθυμητή ιδιότητα:

	A	B	C
1	Short Name	A36	
2			
3	Long Name	Structural Steel A36	
4	Modulus	29000000	
5	Density	0.284	
6	Yield	<code>=VLOOKUP(\$B\$1,\$A\$8:\$E\$12,5)</code>	Yield Strength is in the 5 th column
7			

VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

Παράδειγμα **VLOOKUP**

- Σημειώνεται ότι τα ονόματα δεν είναι case-sensitive (αν είναι κεφαλαία ή μικρά τα γράμματα):

	A	B	C
1	Short Name	ss	
2			
3	Long Name	Stainless Steel 304	
4	Modulus	28,000,000	
5	Density	0.284	
6	Yield Strength	30,000	

Παράδειγμα **VLOOKUP**

- Σε αυτές τις περιπτώσεις, σημειώνεται ότι μια μη-πλήρης ταύτιση με την τιμή που εισάγεται συχνά επιστρέφει μια τιμή:

	A	B	C	D	E
1	Short Name	A37			
2					
3	Long Name	Structural Steel A36			
4	Modulus	29,000,000			
5	Density	0.284			
6	Yield Strength	36,000			
7					
8	Short Name	Long Name	Modulus	Density	Yield Strength
9	A36	Structural Steel A36	29000000	0.284	36000
10	2014	Aluminum 2014-T6	10600000	0.101	60000
11	6061	Aluminum 6061-T6	10000000	0.098	37000
12	SS	Stainless Steel 304	28000000	0.284	30000

	A	B	C	D	E
1	Short Name	2024			
2					
3	Long Name	Aluminum 2014-T6			
4	Modulus	10,600,000			
5	Density	0.101			
6	Yield Strength	60,000			
7					
8	Short Name	Long Name	Modulus	Density	Yield Strength
9	A36	Structural Steel A36	29000000	0.284	36000
10	2014	Aluminum 2014-T6	10600000	0.101	60000
11	6061	Aluminum 6061-T6	10000000	0.098	37000
12	SS	Stainless Steel 304	28000000	0.284	30000

	A	B	C	D	E
1	Short Name	2000			
2					
3	Long Name	#N/A			
4	Modulus	#N/A			
5	Density	#N/A			
6	Yield Strength	#N/A			
7					
8	Short Name	Long Name	Modulus	Density	Yield Strength
9	A36	Structural Steel A36	29000000	0.284	36000
10	2014	Aluminum 2014-T6	10600000	0.101	60000
11	6061	Aluminum 6061-T6	10000000	0.098	37000
12	SS	Stainless Steel 304	28000000	0.284	30000

Παράδειγμα **VLOOKUP**

- Συχνά απαιτούμε πλήρη ταύτιση με την τιμή που ψάχνουμε
- Προκειμένου να γίνει αυτό, πρέπει να οριστεί το τέταρτο όρισμα το οποίο είναι προαιρετικό:
 - TRUE = εύρεση μη-πλήρους ταύτισης
 - FALSE = εύρεση πλήρους ταύτισης

	A	B	C	D
1	Short Name	A37		
2				
3	Long	=VLOOKUP(\$B\$1,\$A\$8:\$E\$12,2,FALSE)		
4	Modulus	25,000,000		
5	Density	0.284		
6	Yield Strength	36,000		

VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

- TRUE - Approximate match
- FALSE - Exact match

Σημείωση σχετικά με τους **Lookup Tables**

- Οι **Lookup tables** δεν πραγματοποιούν γραμμική παρεμβολή!
- Όταν αναζητάται (looking up) μια αριθμητική τιμή και δεν βρίσκεται μια ακριβής ταύτιση με αυτή (και η επιλογή “TRUE” επιτρέπει την μη-πλήρη ταύτιση), τότε η τιμή που αναζητάται στρογγυλοποιείται στην επόμενη προς την επόμενη τιμή του πίνακα.
- Αυτό θα γίνει εμφανές στο παράδειγμα που ακολουθεί

Παράδειγμα με πληθυσμούς

- Τα δεδομένα του πληθυσμού μιας πόλης δίνεται στον διπλανό πίνακα
- Θέλουμε να χρησιμοποιήσουμε ένα **lookup table** ώστε να αναφέρεται ο πληθυσμός όταν εισαχθεί μια συγκεκριμένη χρονιά

	A	B
1	YEAR	POPULATION
2	1956	
3		
4		
5	YEAR	POPULATION
6	1900	550
7	1910	720
8	1920	1256
9	1930	2287
10	1940	2756
11	1950	3012
12	1960	4086
13	1970	5220
14	1980	7478
15	1990	10112
16	2000	14226

Επίλυση με χρήση **Lookup Table**


- Σημειώστε ότι χρησιμοποιώντας **VLOOKUP** συνάρτηση (χωρίς να χρησιμοποιείται η επιλογή TRUE/FALSE ή χρησιμοποιώντας την επιλογή TRUE/FALSE αλλά ορισμένη στο TRUE) επιστρέφει τον πληθυσμό για την επόμενη χρονιά που υπάρχει στον πίνακα, η οποία βρίσκεται χαμηλότερα από τη χρονιά που ζητάται:

	A	B	C
1	YEAR	POPULATION	
2	<code>=vlookup(A2,A5:B16,2)</code>		
3			

	A	B
1	YEAR	POPULATION
2	1956	3012
3		
4		
5	YEAR	POPULATION
6	1900	550
7	1910	720
8	1920	1256
9	1930	2287
10	1940	2756
11	1950	3012
12	1960	4086
13	1970	5220
14	1980	7478
15	1990	10112
16	2000	14226


Γραμμική παρεμβολή (Linear Interpolation)

- Μπορούμε να χρησιμοποιήσουμε τις δηλώσεις **IF** προκειμένου να πάρουμε μια καλύτερη προσέγγιση χρησιμοποιώντας **γραμμική προσαρμογή (linear interpolation)**
- Στη γραμμική προσαρμογή, υποθέτουμε ότι τα δεδομένα μας **μεταβάλλονται γραμμικά** μεταξύ των δεδομένων (data points):




Εξίσωση για γραμμική παρεμβολή (Linear Interpolation)

- Έστω ότι το σημείο x βρίσκεται μεταξύ των x_1 και x_2 . Οι τιμές του y οι οποίες αντιστοιχούν στο x_1 και στο x_2 είναι δεδομένες (y_1 και y_2 , αντίστοιχα). Θέλουμε να βρούμε την τιμή του y που αντιστοιχεί στο x .



Εξίσωση για γραμμική παρεμβολή (Linear Interpolation)

- Όμοια τρίγωνα:




$$\frac{y - y_1}{x - x_1} = \frac{y_2 - y_1}{x_2 - x_1}$$

Εξίσωση για γραμμική παρεμβολή (Linear Interpolation)

$$\frac{y - y_1}{x - x_1} = \frac{y_2 - y_1}{x_2 - x_1}$$

$$y = y_1 + \left(\frac{y_2 - y_1}{x_2 - x_1} \right) (x - x_1)$$

Παράδειγμα πληθυσμού



$$y = 3012 + \left(\frac{4086 - 3012}{1960 - 1950}\right)(1956 - 1950) = 3656$$

Γραμμική προσαρμογή σε ένα πίνακα του Excel

- Για κάθε διάστημα που μεσολαβεί μεταξύ δύο διαδοχικών τιμών του πληθυσμού, κατασκευάζουμε μια δήλωση **IF** για να δούμε αν η χρονιά που μας ζητήθηκε πέφτει εντός του διαστήματος...

	A	B	C	D	E
1	YEAR	POPULATION			
2	1956				
3					
4					
5	YEAR	POPULATION			
6	1900	550			
7	1910	720	=if(and(A2>=A6,A2<=A7),		
8	1920	1256	IF(logical_test, [value_if_true], [value_if_false])		

Γραμμική προσαρμογή σε ένα πίνακα του Excel

- Εάν η χρονιά που μας ζητάνε πέφτει εντός του διαστήματος, τότε πρέπει να χρησιμοποιηθεί η εξίσωση της γραμμικής παρεμβολής...

	A	B	C	D	E	F	G
1	YEAR	POPULATION					
2	1956						
3							
4							
5	YEAR	POPULATION					
6	1900	550					
7	1910	720	=if(and(A2>=A6,A2<=A7),B6+(B7-B6)/(A7-A6)*(A2-A6),				
8	1920	1256					

Do if condition is true

IF(logical_test, [value_if_true], [value_if_false])

Γραμμική προσαρμογή σε ένα πίνακα του Excel

- Εάν η χρονιά που μας ζητάνε δεν είναι εντός του διαστήματος, τότε εισάγεται ένας κενός χαρακτήρας (space) στο συγκεκριμένο κελί, ώστε να μείνει κενό(blank).
- Κλειδώνουμε τις συντεταγμένες του κελιού (A2) ώστε να επιτραπεί αντιγραφή της εξίσωσης σωστά σε επόμενα κελιά

	A	B	C	D	E	F	G	H
1	YEAR	POPULATION						
2	1956							
3								
4								
5	YEAR	POPULATION						
6	1900	550						
7	1910	720	=if(and(\$A\$2>=A6,\$A\$2<=A7),B6+(B7-B6)/(A7-A6)*(\$A\$2-A6)," ")					

Γραμμική προσαρμογή σε ένα πίνακα του Excel

- Αντιγράφουμε την εξίσωση
- Το μόνο κελί, το οποίο έχει κάποια τιμή θα είναι αυτό που αντιστοιχεί στο διάστημα που περιέχει τη χρονιά που μας ζητήθηκε
- **Σημείωση:** η εισαγωγή του space στα υπόλοιπα κελιά είναι προτιμότερη από την εισαγωγή του μηδενός, επειδή μερικοί πίνακες θα έχουν θετικές και αρνητικές τιμές.

YEAR	POPULATION	
1956		
YEAR	POPULATION	
1900	550	
1910	720	
1920	1256	
1930	2287	
1940	2756	
1950	3012	
1960	4086	3656.4
1970	5220	
1980	7478	
1990	10112	
2000	14226	

Γραμμική προσαρμογή σε ένα πίνακα του Excel

- Ο πληθυσμός βρίσκεται παίρνοντας τη μέγιστη τιμή (τη μόνη αριθμητική τιμή) στο διάστημα των υπολογισμένων τιμών

	A	B	C	
1	YEAR	POPULATION		
2	1956	=MAX(C7:C16)		
3		MAX(number1, [number2], ...)		
4				
5	YEAR	POPULATION		
6	1900	550		
7	1910	720		
8	1920	1256		
9	1930	2287		
10	1940	2756		
11	1950	3012		
12	1960	4086	3656.4	
13	1970	5220		
14	1980	7478		
15	1990	10112		
16	2000	14226		

Γραμμική προσαρμογή σε ένα πίνακα του Excel

- Εάν είναι επιθυμητό, η εξίσωση του πληθυσμού μπορεί να τροποποιηθεί ώστε να αναφέρει **σφάλμα** όταν η τιμή της χρονιάς που εισάγεται δεν πέφτει εντός του εύρους του πίνακα

	A	B	C	D	E	F
1	YEAR	POPULATION				
2	1956	=IF(AND(A2>=A6,A2<=A16),MAX(C7:C16),"Out of range")				
3						
4						
5	YEAR	POPULATION				
6	1900	550				
7	1910	720				
8	1920	1256				
9	1930	2287				
10	1940	2756				
11	1950	3012				
12	1960	4086	3656.4			
13	1970	5220				
14	1980	7478				
15	1990	10112				
16	2000	14226				

Έλεγχοι για διάφορες τιμές - παραδείγματα

	A	B
1	YEAR	POPULATION
2	2007	Out of range
3		
4		
5	YEAR	POPULATION
6	1900	550
7	1910	720
8	1920	1256
9	1930	2287
10	1940	2756
11	1950	3012
12	1960	4086
13	1970	5220
14	1980	7478
15	1990	10112
16	2000	14226

	A	B	C
1	YEAR	POPULATION	
2	1970	5220	
3			
4			
5	YEAR	POPULATION	
6	1900	550	
7	1910	720	
8	1920	1256	
9	1930	2287	
10	1940	2756	
11	1950	3012	
12	1960	4086	
13	1970	5220	5220
14	1980	7478	5220
15	1990	10112	
16	2000	14226	

η χρονιά 1970 πέφτει εντός δύο διαστημάτων, αλλά και τα δύο δίνουν τελικά την ίδια τιμή

Έλεγχοι για διάφορες τιμές - παραδείγματα

	A	B	C
1	YEAR	POPULATION	
2	1988	9585	
3			
4			
5	YEAR	POPULATION	
6	1900	550	
7	1910	720	
8	1920	1256	
9	1930	2287	
10	1940	2756	
11	1950	3012	
12	1960	4086	
13	1970	5220	
14	1980	7478	
15	1990	10112	9585.2
16	2000	14226	

	A	B	C
1	YEAR	POPULATION	
2	1900	550	
3			
4			
5	YEAR	POPULATION	
6	1900	550	
7	1910	720	550
8	1920	1256	
9	1930	2287	
10	1940	2756	
11	1950	3012	
12	1960	4086	
13	1970	5220	
14	1980	7478	
15	1990	10112	
16	2000	14226	

	A	B	
1	YEAR	POPULATION	
2	1899	Out of range	
3			
4			
5	YEAR	POPULATION	
6	1900	550	
7	1910	720	
8	1920	1256	
9	1930	2287	
10	1940	2756	
11	1950	3012	
12	1960	4086	
13	1970	5220	
14	1980	7478	
15	1990	10112	
16	2000	14226	

Συμπεράσματα

- Οι **δηλώσεις IF** προσφέρουν μεγάλη ευελιξία στα Excel spreadsheets – επιτρέπουν έλεγχο στο ποιοι υπολογισμοί θα πραγματοποιηθούν
- Οι **Lookup tables** είναι βολικοί όταν επιθυμούμε πρόσβαση σε δεδομένα, τα οποία είναι αποθηκευμένα σε πίνακα. Σημαντικό είναι να θυμόμαστε ότι οι lookup tables δεν κάνουν παρεμβολή μεταξύ τιμών
- Οι **εξισώσεις γραμμικής παρεμβολής** μπορούν να προστεθούν σε πίνακες προκειμένου να παρέχουν μια καλύτερη εκτίμηση των τιμών μεταξύ των δεδομένων του πίνακα

Χρηματοδότηση

- Το παρόν εκπαιδευτικό υλικό έχει αναπτυχθεί στο πλαίσιο του εκπαιδευτικού έργου του διδάσκοντα.
- Το έργο «**Ανοικτά Ακαδημαϊκά Μαθήματα στο Πανεπιστήμιο Πατρών**» έχει χρηματοδοτήσει μόνο την αναδιαμόρφωση του εκπαιδευτικού υλικού.
- Το έργο υλοποιείται στο πλαίσιο του Επιχειρησιακού Προγράμματος «Εκπαίδευση και Δια Βίου Μάθηση» και συγχρηματοδοτείται από την Ευρωπαϊκή Ένωση (Ευρωπαϊκό Κοινωνικό Ταμείο) και από εθνικούς πόρους.



Σημείωμα Αναφοράς

Copyright©2014 Πανεπιστήμιο Πατρών, Τμήμα Χημικών Μηχανικών,
Καθ. Δημήτρης Ματαράς (mataras@upatras.gr), «Εργαστήριο
Υπολογιστών» «Logical IF Statements, Lookup Tables, Linear
Interpolation»

Διαθέσιμο από τη δικτυακή διεύθυνση:

<https://eclass.upatras.gr/courses/CMNG2112/>

Σημείωμα Αδειοδότησης

Το παρόν υλικό διατίθεται με τους όρους της άδειας χρήσης Creative Commons Αναφορά, Μη Εμπορική Χρήση Παρόμοια Διανομή 4.0 [1] ή μεταγενέστερη, Διεθνής Έκδοση. Εξαιρούνται τα αυτοτελή έργα τρίτων π.χ. φωτογραφίες, διαγράμματα κ.λ.π., τα οποία εμπεριέχονται σε αυτό και τα οποία αναφέρονται μαζί με τους όρους χρήσης τους στο «Σημείωμα Χρήσης Έργων Τρίτων».



[1] <http://creativecommons.org/licenses/by-nc-sa/4.0/>

Ως **Μη Εμπορική** ορίζεται η χρήση:

- που δεν περιλαμβάνει άμεσο ή έμμεσο οικονομικό όφελος από την χρήση του έργου, για το διανομέα του έργου και αδειοδόχο
- που δεν περιλαμβάνει οικονομική συναλλαγή ως προϋπόθεση για τη χρήση ή πρόσβαση στο έργο
- που δεν προσπορίζει στο διανομέα του έργου και αδειοδόχο έμμεσο οικονομικό όφελος (π.χ. διαφημίσεις) από την προβολή του έργου σε διαδικτυακό τόπο

Ο δικαιούχος μπορεί να παρέχει στον αδειοδόχο ξεχωριστή άδεια να χρησιμοποιεί το έργο για εμπορική χρήση, εφόσον αυτό του ζητηθεί.